Review Various Methods to Perform the Analysis of a 2 Treatment, 2 Period Crossover Study

By
Rocco Brunelle
IPT

SPROM - March 1, 2000

Crossover Study Design



Grizzle Model

$Model:\ Y_{ijk} = m + b_{ij} + p_k + f_m + l_m + e_{ijk}$

where $u = overall\ mean$

$b_{ij} = effect\ of\ j^{th}\ patient\ within\ i^{th}\ sequence\ and\ is\ N(0, s_b^2)$

$p_k = effect\ of\ k^{th}\ period$

$f_m = direct\ effect\ of\ m^{th}\ drug$

$l_m = residual\ effect\ of\ m^{th}\ drug$

$e_{ijk} = random\ error\ and\ is\ N(0, s_e^2)$

$with\ \ Var(Y_{ijk}) = s_b^2 + s_e^2\ \ and\ \ Cov(Y_{ijk}) = s_b^2$

Cell Means Model

$Model:\ Y_{ijk} = m + b_{ij} + p_k + f_m + l_m + e_{ijk}$

Where i=sequence, j=patient, k=period and m=treatment

	Period 1	Period 2
Seq. 1	$m + p_1 + f_1\ (\bar{Y}_{1.1})$	$m + p_2 + f_2 + l_1\ (\bar{Y}_{2.1})$
Seq. 2	$m + p_1 + f_2\ (\bar{Y}_{1.2})$	$m + p_2 + f_1 + l_2\ (\bar{Y}_{2.2})$

Where (sequence,patient,period)

Estimating Model Effects

Estimation of Carryover Effects

$$l_1 - l_2 = (\bar{Y}_{1.1} + \bar{Y}_{2.1}) - (\bar{Y}_{1.2} + \bar{Y}_{2.2})$$

Seq. AB Seq. BA

Where (sequence,patient,period)

Estimating Model Effects

Estimation of Treatment Effect

$$f_1 - f_2 = \bar{Y}_{1.1} - \bar{Y}_{1.2}$$

using first period data

$$f_1 - f_2 = \{(\bar{Y}_{1.1} - \bar{Y}_{2.1}) - (\bar{Y}_{1.2} - \bar{Y}_{2.2})\}/2$$

assuming $l_1 = l_2$

Where (sequence, patient, period)

Grizzle's Model

Using GLM.

```
PROC GLM DATA=ORIG ;
     CLASS SEQ PATIENT PERIOD TRT ;
     MODEL RESULT = SEQ PATIENT(SEQ) PERIOD TRT ;
     RANDOM PATIENT(SEQ) ;
     TEST H=SEQ E=PATIENT(SEQ) ;
     LSMEANS SEQ TRT ;
RUN ;
```

Use Type III Sums of Squares

Grizzle's Model

Using Proc Mixed with a Random Statement.

```
PROC MIXED DATA=ORIG ;
     CLASS SEQ PATIENT PERIOD TRT ;
     MODEL RESULT = SEQ PERIOD TRT ;
     RANDOM PATIENT(SEQ) / SUBJECT=PATIENT
                TYPE=CS;
RUN ;
```

(Thanks to Jie Huang for the Proc Mixed Code)

Grizzle's Model

Using Proc Mixed with a Repeated Statement.

```
PROC MIXED DATA=ORIG ;
     CLASS SEQ PATIENT PERIOD TRT ;
     MODEL RESULT = SEQ PERIOD TRT ;
     REPEATED  / SUBJECT=PATIENT  TYPE=CS R;
RUN ;
```

Obtained the same results with Type=UN.

Crossover Analyses

- Step 1- Test for carry-over effect (α=.10)
- Step 2
  - If the carry-over effect is significant than only Period 1 data can estimate the treatment effect.
  - If the carry-over effect is not significant than the pooled data from both periods can be used to estimate the treatment effect

No Carryover Effect



Carryover Effect

Physical Carryover Effect



Rebound Phenomenon



Carryover Effect



Crossover Design Assumptions

- Chronic condition
- Relatively Short Study Period
- No Carryover Effects Expected
  - Clinical and scientific rational for no carryover effects before the study is started. Also, perform an analysis to test for potential carryover effects.

Steps in Performing a Crossover Analysis

- Step 1 - Analysis by Period
  - Descriptive statistics by visit
  - Graphics by visit
  - Analysis comparing treatments by visit
- Step 2 - Pooled Analysis
  - Descriptive statistics by visit
  - Graphics by visit
  - Crossover model testing carryover effect and treatment effect

Example

- 2 hr post prandial BG for study IOAG
- Analyses by Period
  - Descriptive statistics and analyses
  - Graphs
- Analyses for Combined Periods
  - Descriptive statistics and analyses
  - Graphs

Koch's Analysis for a Crossover Design

Patient	Sequence	Period 1	Period 2	Sum	Difference	Crossover Difference
1	AB	$X_{1,1}$	$X_{2,1}$	$X_{1,1}+ X_{2,1}$	$X_{1,1}- X_{2,1}$	$X_{1,1}- X_{2,1}$
2	BA	$X_{1,2}$	$X_{2,2}$	$X_{1,2}+ X_{2,2}$	$X_{1,2}- X_{2,2}$	$X_{2,2}- X_{1,2}$

$H_o:\lambda_1=\lambda_2$ Compare sequence groups using Sums
$H_o:\phi_1=\phi_2$ Compare sequence groups using Differences

Non-parametric analyses by ranking the sums or differences and then performing the analyses on the ranks.

Miscellaneous Issues

- Difference between analyzing actual measurements or change from baseline
- Using the rank transformation
  - Grizzle's Model
  - Koch's Model
- Missing data
  - Grizzle's Model
  - Koch's Model

```
title 'Crossover Analysis Example' ;
title2 'Compares GLM, Mixed and Kochs Methods' ;

options mtrace mprint ;
options pageno=1 ps=130 ls=53 ;
footnote1 'pagebreak' ;

title3 'ORIG1: Crossover with only a treatment effect' ;

data orig1 ;
     length seq trt $2 ;
     input pat seq period trt result ;
cards ;
1  AB 1 A 15
1  AB 2 B 10
2  AB 1 A 14
2  AB 2 B 11
3  AB 1 A 16
3  AB 2 B 10
4  AB 1 A 15
4  AB 2 B  9
5  AB 1 A 16
5  AB 2 B 09
6  BA 1 B 11
6  BA 2 A 15
7  BA 1 B 10
7  BA 2 A 14
8  BA 1 B 09
8  BA 2 A 16
9  BA 1 B 12
9  BA 2 A 15
10 BA 1 B 11
10 BA 2 A 14
run ;

%macro xover(data= ,title= ) ;

proc dataset ;
     delete _koch ;
run ;

title3 "Data=&data : &title" ;

proc sort data=&data ; by period trt ;
run ;

proc means data=&data ;
     by period trt ;
     var result ;
run ;

* ***************************** *;
* GLM                           *;
* ***************************** *;
 /*
title4 'Grizzles Model Using GML' ;
proc glm data=&data ;
```

```
     classes pat seq period trt ;
     model result=seq pat(seq) period trt / ss1 ss3 ;
     test h=seq e=pat(seq) ;
     lsmean seq trt;
run ;
 */
title4 'Grizzles Model Using GML with a Random Statement' ;
proc glm data=&data ;
     classes pat seq period trt ;
     model result=seq pat(seq) period trt / ss1 ss3 ;
     random pat(seq);
     test h=seq e=pat(seq) ;
     lsmean seq trt;
run ;

* ***************************** *;
* Mixed-1, random statement     *;
* ***************************** *;

proc sort data=&data ; by trt period ;
run ;

title4 'Grizzles Model Using Proc Mixed with Random Statement' ;
proc mixed data=&data;
     class seq pat trt period;
     model result=seq period trt ;
     random pat(seq) /subject=pat type=cs;
run;

* ***************************** *;
* Mixed-2, repeat statement     *;
* ***************************** *;

title4 'Grizzles Model Using Proc Mixed with Repeated Statement' ;
title5 'Type=CS' ;
proc mixed data=&data;
     class seq trt period pat;
     model result=seq period trt ;
     repeated /type=cs sub=pat(seq) r;
run;

title5 'Type=UN' ;
proc mixed data=&data;
     class seq trt period pat;
     model result=seq period trt ;
     repeated /type=un sub=pat(seq) r;
run;

proc sort data=&data ; by pat seq period trt ;
run ;

proc transpose data=&data out=_koch prefix=period  ; by pat seq ;
     var result ;
     id period ;
run ;

data _koch ; set _koch ;
```

```
     sum=period1+period2 ;
     dif=period1-period2 ;
     if seq='AB' then dif2=period1-period2 ;
     else dif2=period2-period1 ;
run ;

proc print data=&data ;
run ;
proc print data=_koch ;
run ;

title4 'Kochs Model' ;
proc ttest data=_koch ;
     class seq ;
     var sum dif dif2 ;
run ;

%mend xover ;

%xover(data=orig1, title=Crossover Model with Treatment Effect) ;
```

Analysis Variable : RESULT

PERIOD=1 TRT=A

N	Mean	Std Dev	Minimum	Maximum
5	15.2000000	0.8366600	14.0000000	16.0000000

PERIOD=1 TRT=B

N	Mean	Std Dev	Minimum	Maximum
5	10.6000000	1.1401754	9.0000000	12.0000000

PERIOD=2 TRT=A

N	Mean	Std Dev	Minimum	Maximum
5	14.8000000	0.8366600	14.0000000	16.0000000

PERIOD=2 TRT=B

N	Mean	Std Dev	Minimum	Maximum
5	9.8000000	0.8366600	9.0000000	11.0000000

General Linear Models Procedure
Class Level Information

Class	Levels	Values
PAT	10	1 2 3 4 5 6 7 8 9 10
SEQ	2	AB BA
PERIOD	2	1 2
TRT	2	A B

Number of observations in data set = 20

General Linear Models Procedure

Dependent Variable: RESULT

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	11	120.80000000	10.98181818	8.79	0.0024
Error	8	10.00000000	1.25000000		
Corrected Total	19	130.80000000			

R-Square	C.V.	Root MSE	RESULT Mean
0.923547	8.873286	1.11803399	12.60000000

Source	DF	Type I SS	Mean Square	F Value	Pr > F
SEQ	1	0.20000000	0.20000000	0.16	0.6996
PAT(SEQ)	8	3.60000000	0.45000000	0.36	0.9151
PERIOD	1	1.80000000	1.80000000	1.44	0.2645
TRT	1	115.20000000	115.20000000	92.16	0.0001

Source	DF	Type III SS	Mean Square	F Value	Pr > F
SEQ	1	0.20000000	0.20000000	0.16	0.6996
PAT(SEQ)	8	3.60000000	0.45000000	0.36	0.9151
PERIOD	1	1.80000000	1.80000000	1.44	0.2645
TRT	1	115.20000000	115.20000000	92.16	0.0001

General Linear Models Procedure

Source	Type III Expected Mean Square
SEQ	Var(Error) + 2 Var(PAT(SEQ)) + Q(SEQ)
PAT(SEQ)	Var(Error) + 2 Var(PAT(SEQ))
PERIOD	Var(Error) + Q(PERIOD)
TRT	Var(Error) + Q(TRT)

General Linear Models Procedure
Least Squares Means

SEQ	RESULT LSMEAN
AB	12.5000000
BA	12.7000000

TRT	RESULT LSMEAN
A	15.0000000
B	10.2000000

General Linear Models Procedure

Dependent Variable: RESULT

Tests of Hypotheses using the Type III MS for PAT(SEQ) as an error term

Source	DF	Type III SS	Mean Square	F Value	Pr > F
SEQ	1	0.20000000	0.20000000	0.44	0.5237

The MIXED Procedure

Class Level Information

Class	Levels	Values
SEQ	2	AB BA
PAT	10	1 2 3 4 5 6 7 8 9 10
TRT	2	A B
PERIOD	2	1 2

REML Estimation Iteration History

Iteration	Evaluations	Objective	Criterion
0	1	21.22374314	
1	2	19.35964672	.
2	1	19.23627125	0.46206526
3	1	19.22135696	0.00588083
4	1	19.22113291	0.00000157
5	1	19.22113285	0.00000000

Convergence criteria met.

Covariance Parameter Estimates (REML)

Cov Parm	Subject	Estimate
Variance	PAT	0.02606382
CS	PAT	-0.42606385
Residual		1.25000004

Model Fitting Information for RESULT

Description	Value
Observations	20.0000
Res Log Likelihood	-24.3136
Akaike's Information Criterion	-27.3136
Schwarz's Bayesian Criterion	-28.4725
-2 Res Log Likelihood	48.6272
Null Model LRT Chi-Square	2.0026
Null Model LRT DF	2.0000
Null Model LRT P-Value	0.3674

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
SEQ	1	8	0.44	0.5237
PERIOD	1	8	1.44	0.2645
TRT	1	8	92.16	0.0001

The MIXED Procedure

Class Level Information

Class	Levels	Values
SEQ	2	AB BA
TRT	2	A B
PERIOD	2	1 2
PAT	10	1 2 3 4 5 6 7 8 9 10

REML Estimation Iteration History

Iteration	Evaluations	Objective	Criterion
0	1	21.22374314	
1	1	19.22113285	0.00000000

Convergence criteria met.

R Matrix for PAT(SEQ) 1 AB

Row	COL1	COL2
1	0.85000000	-0.40000000
2	-0.40000000	0.85000000

Covariance Parameter Estimates (REML)

Cov Parm	Subject	Estimate
CS	PAT(SEQ)	-0.40000000
Residual		1.25000000

Model Fitting Information for RESULT

Description	Value
Observations	20.0000
Res Log Likelihood	-24.3136
Akaike's Information Criterion	-26.3136
Schwarz's Bayesian Criterion	-27.0862
-2 Res Log Likelihood	48.6272
Null Model LRT Chi-Square	2.0026
Null Model LRT DF	1.0000
Null Model LRT P-Value	0.1570

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
SEQ	1	8	0.44	0.5237
PERIOD	1	8	1.44	0.2645
TRT	1	8	92.16	0.0001

The MIXED Procedure

Class Level Information

Class	Levels	Values
SEQ	2	AB BA
TRT	2	A B
PERIOD	2	1 2
PAT	10	1 2 3 4 5 6 7 8 9 10

REML Estimation Iteration History

Iteration	Evaluations	Objective	Criterion
0	1	21.22374314	
1	1	18.89455690	0.00000000

Convergence criteria met.

R Matrix for PAT(SEQ) 1 AB

Row	COL1	COL2
1	0.70000000	-0.40000000
2	-0.40000000	1.00000000

Covariance Parameter Estimates (REML)

Cov Parm	Subject	Estimate
UN(1,1)	PAT(SEQ)	0.70000000
UN(2,1)	PAT(SEQ)	-0.40000000
UN(2,2)	PAT(SEQ)	1.00000000

Model Fitting Information for RESULT

Description	Value
Observations	20.0000
Res Log Likelihood	-24.1503
Akaike's Information Criterion	-27.1503
Schwarz's Bayesian Criterion	-28.3092
-2 Res Log Likelihood	48.3006
Null Model LRT Chi-Square	2.3292
Null Model LRT DF	2.0000
Null Model LRT P-Value	0.3120

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
SEQ	1	8	0.44	0.5237
PERIOD	1	8	1.44	0.2645
TRT	1	8	92.16	0.0001

OBS	SEQ	TRT	PAT	PERIOD	RESULT
1	AB	A	1	1	15
2	AB	B	1	2	10
3	AB	A	2	1	14
4	AB	B	2	2	11
5	AB	A	3	1	16
6	AB	B	3	2	10
7	AB	A	4	1	15
8	AB	B	4	2	9
9	AB	A	5	1	16
10	AB	B	5	2	9
11	BA	B	6	1	11
12	BA	A	6	2	15
13	BA	B	7	1	10
14	BA	A	7	2	14
15	BA	B	8	1	9
16	BA	A	8	2	16
17	BA	B	9	1	12
18	BA	A	9	2	15
19	BA	B	10	1	11
20	BA	A	10	2	14

OBS	PAT	SEQ	_NAME_	PERIOD1	PERIOD2	SUM	DIF	DIF2
1	1	AB	RESULT	15	10	25	5	5
2	2	AB	RESULT	14	11	25	3	3
3	3	AB	RESULT	16	10	26	6	6
4	4	AB	RESULT	15	9	24	6	6
5	5	AB	RESULT	16	9	25	7	7
6	6	BA	RESULT	11	15	26	-4	4
7	7	BA	RESULT	10	14	24	-4	4
8	8	BA	RESULT	9	16	25	-7	7
9	9	BA	RESULT	12	15	27	-3	3
10	10	BA	RESULT	11	14	25	-3	3

TTEST PROCEDURE

Variable: SUM

SEQ	N	Mean	Std Dev	Std Error	Variances	T	DF	Prob>|T|
AB	5	25.00000000	0.70710678	0.31622777	Unequal	-0.6667	6.7	0.5274
BA	5	25.40000000	1.14017543	0.50990195	Equal	-0.6667	8.0	0.5237

For H0: Variances are equal, F' = 2.60 DF = (4,4) Prob>F' = 0.3772

****

Variable: DIF

SEQ	N	Mean	Std Dev	Std Error	Variances	T	DF	Prob>|T|
AB	5	5.40000000	1.51657509	0.67823300	Unequal	9.6000	7.9	0.0001
BA	5	-4.20000000	1.64316767	0.73484692	Equal	9.6000	8.0	0.0000

For H0: Variances are equal, F' = 1.17 DF = (4,4) Prob>F' = 0.8803

****

Variable: DIF2

SEQ	N	Mean	Std Dev	Std Error	Variances	T	DF	Prob>|T|
AB	5	5.40000000	1.51657509	0.67823300	Unequal	1.2000	7.9	0.2647
BA	5	4.20000000	1.64316767	0.73484692	Equal	1.2000	8.0	0.2645

For H0: Variances are equal, F' = 1.17 DF = (4,4) Prob>F' = 0.8803